

“Teen World”

New Mexico Adventures in Supercomputing Challenge

Final Report

April 7th, 2004

Team number 95

Monte Del Sol Charter School

Santa Fe, New Mexico

Team members:

Caroline Critchlow

Sonja Romero

Sponsoring teacher:

Paige Prescott

Project mentor:

Irene Lee

Table of Contents:

Executive summary	3
Introduction	4
The Agent specification	4
The Environment	6
The User Interface	7
Running the Simulation	7
The Results	9
The Conclusions	10
References	11
Acknowledgements	12
Appendix A: The Code	13
Appendix B: The Charts	18

Executive Summary

Since 1998 New Mexico suicide rates have been slowly on the rise with 294 suicides in 1998 and 309 suicides in 2000 (latest available statistics). The total number of suicides in the United States for 2002 was 29,350.

The New Mexico suicide rates are currently the second highest in the nation, following Alaska, with 19.1 deaths per 100,000 teens or roughly 1 in every 5000 teens. In 2002, 26 New Mexico kids committed suicide between the ages of 10-19 including two in Santa Fe County. For us, these are shocking statistics. In our extended peer group, most of us have had to deal with suicidal thoughts, suicide attempts, and even with suicide itself.

When people take their own lives, it is to stop the pain and hurt that they feel and to end all the sorrow in their lives. What few people stop to realize is that with the suicide of a friend or loved one, the same pain the victim felt spreads through the community of those who knew the victim. This spreading of pain causes depression levels to grow which may result in further suicides.

We have realized this but we doubt that the suicide victims understood this before they became a statistic. With this project, we wish to demonstrate how quickly depression might spread, and how this depression may lead to further suicides. We investigated this scenario both with and without interventions. We also investigated how certain variables and circumstances might factor in to suicide and depression.

Our project, "Teen World", is a simulation of how depression and suicide affect a community. Our simulation uses Starlogo to model the community of teens and see how they respond to events as the simulation progresses. Variables such as emotional state, moodiness, gender, network of friends, density of counselors, and rate of occurrence of negative events will have impact on the teens.

Introduction:

Teen World is a Starlogo simulation of a community of teens. Using Starlogo we are able to create a community of teenagers as agents (called turtles in Starlogo lingo) and counselor as agents. The teenagers live and cope with depression, suicides, and randomly occurring natural deaths, and the counselors counsel depressed teens. Both of these agents operate in a virtual physical environment called the canvas. The canvas consists of a grid of randomly placed neutral, happy and sad spaces that affect a teen's mood when they are encountered. It is also where random events that influence a teen's mood occur sporadically during the simulation. The significance of these elements is to help us see how depression and suicide spread and also what may factor into a suicide.

Agent specifications:

To create this community, there are two types of agents we have to code: teens and counselors.

Teens:

The agents that represent the teens are turtle shaped. A teen agent has variables that are specific to each: emotion-level, moodiness, color, group of friends, and gender.

The emotion-level represents how happy or depressed a teen is. If the emotion-level is a positive value, then the teen is happy (1, being a little happy to 45, being ecstatic). If it is a negative number, then the teen is depressed (-1, being a little sad, to -45, being suicidally depressed). If the emotion-level is zero, the teen is in a neutral state. The magnitude of the value corresponds with the level of happiness or depression. Initially, the emotion-level values are set to zero. The emotion-level of teens varies depending on what they encounter in the simulation.

Moodiness is a randomly assigned integer value between 1 and 10. The moodiness value is specific to each teen and never changes throughout the simulation. When an event happens, then the moodiness value is multiplied by an "event-value" then added to the current emotion-level to get the new emotion-level.

Each teen is randomly assigned to a group of friends that is shown by color or hue. There are five groups of friends in all. Each group of friends is distinct (non-overlapping) and has a different color. The group of friends a teen is in also affects the teen's emotion-level in the event of a suicide or death of a teen within the same group.

Shades of color within a hue represent the emotion-level of the teen. As the emotion-level changes, the color will scale into shades of the same color to correspond with the current emotion-level. For example, if a teen belongs to the "green" friend group and is happy, then the teen's color will be a lighter shade of green. If the same turtle becomes depressed, then it will turn a darker shade of green.

Gender is set to help us make our simulation more realistic. In suicide studies it has been seen that boys' suicide attempts have a higher success rate than girls' suicide attempts because more male suicides are committed using firearms. Girls however, are more likely to slit their wrists, a less successful way of attempting suicide. Statistics show that every 1 in 5 boys successfully commit suicide whereas only 1 in 20 girls will successfully commit suicide.

Teen actions:

Teens move by turning in a random direction and taking a step forward. From any given patch a teen moves to one of the 8 surrounding patches that make up its "neighborhood". When a teen reaches the edge of the canvas and moves another step away from the center of the canvas, the teen ends up on the opposite edge. For example if a teen walks off the top of the canvas, they end up on the bottom of the canvas and if a teen walks off the right side of the canvas, they end up on the left side of the canvas. Instead of turning around or bouncing off the "wall", the teen will appear on the opposite side of the canvas.

When the teens move around, they react to the patches they encounter. With each step, the teen's emotion-level can rise or fall due to the patch they encounter.

$$emotion-level = emotion-level + (event-value * moodiness)$$

where event-value is:

-0.5 if a sad space is encountered

0.5 if a happy space is encountered

-0.3, -0.6, or -1 if the site of an accidental death is encountered
(randomly selected from these three values)

-0.5 or -1 if the site of a suicide is encountered
(randomly selected from these two values)

At each step a teen has the possibility of dying an accidental death. Every teen randomly selects a number between zero and the "odds of accidental death" - 1. If the result is 1 then the teen suffers an accidental death.

At each step a teen may be affected by the death of a friend.

if a friend dies an accidental death,

$$emotion-level = emotion-level + (-0.75 * moodiness)$$

if a friend commits suicide,

$$emotion-level = emotion-level + (-1.0 * moodiness)$$

(We set the teens' emotion-level to be lowered more by a suicide than by an accidental death because we read that the suicide of a friend has greater impact than an accidental death. The friends of a suicide victim tend to blame themselves for the suicide or feel that the suicide may have been preventable thus resulting in greater depression than if the death were accidental.)

At each step if a teen becomes suicidal (emotion-level = -45) then he/she will attempt suicide. The rate of success in committing suicide varies by gender. This is reflected in our code.

```

ifelse gender = 0                                ; if gender is male
  [if (random 5) = 0 [suicide]]                  ; 1 in 5 times the result is a suicide
  [if (random 20) = 0 [suicide]]                 ; else if gender is female
                                                ; 1 in 20 times the result is a suicide

```

Counselors:

The counselor agents are a special shape that resembles the nurse's cross in green within a white circle. The number of teens and the density of counselors specified by the user determine the number of counselors in the simulation.

$$\# \text{ of counselors} = \# \text{ of teens} * (\text{density of counselors} / 100)$$

Statistics show that there are 500 teens in need of counseling for every 1 counselor and that 1 in every 20 teens needs counseling.

The counselor's sole function is to walk around and check the surrounding neighborhood of 8 patches for a teen with a negative emotion-level. Upon finding a depressed teen, the counselor changes the teen's emotion-level. One half of the time the counseling has a positive effect on the teen, raising the teen's emotion-level, one quarter of the time counseling has no effect, and one quarter of the time the counseling actually decreases the teen's emotion level. (This assumption was based on personal experience.)

$$\text{emotion-level} = \text{emotion-level} + (((\text{random } 4) - 1) * \text{moodiness})$$

; where (random 4) returns 0, 1, 2, or 3

The Environment:

The simulation environment or canvas consists of patches arranged in a grid. The StarLogo environment is a wrap-around canvas, in which turtles can wander off edges and appear on opposite edges.

Patches are used to represent physical events that affect emotion-level. Initially all patches are black or neutral areas like streets or hallways where any random event may happen such as a car accident or a surprise party that affects the teens' emotion-level. We are using four other colors of patches: dark gray, white, purple, and pink. Dark gray

patches represent “bad or depressing zones” where the turtles become more depressed. White patches are “good or happy zones” where the teens become happier. The dark gray and white patches are randomly distributed during the setup phase of the program. Purple patches mark where a person has committed suicide. Pink patches mark where a person has died accidentally. Purple and pink patches arise during the actually running of the program.

The User Interface:

Before starting the program, certain variables can be set by the user.

- The “**# of Teens**” slider controls the number of teens to be created. This variable might help us see how teens might react in bigger groups vs. smaller groups.
- The “**odds of accidental death**” slider determines the rate of accidental death among teens. Each teen has a 1 in # chance of death each step they take.
- The “**density of counselors**” slider determines how many counselors exist in the simulation as compared to the number of teens.
- The “**% of happy zones**” slider controls the number of white of happy patches that are randomly distributed.*
- The “**% of sad zones**” slider controls the number of dark gray or sad patches that are randomly distributed.*

* the sum of the % of happy zones and % of sad zones slider values cannot equal more than 100%

Once these variables are set, the user uses the setup button and go button to actually run the program.

Running the Simulation

Setup phase:

During the setup phase (that is kicked off when the setup button is pressed), a single turtle is created to jump around and color patches either gray or white representing sad and happy places. The turtle only colors patches if they are black. Upon completing this task, the turtle will die so it will not interfere with the rest of the simulation. Next, the “# of teens” slider value is used to control the number of teens created, and their variables are set including the x-y position in the environment. Then the “density of counselors” slider value is used to create the counselors and they are placed randomly in the environment.

Go loop:

Once the setup phase is completed, then the user must push the “go” button. Then the turtles wander around. If they walk over a dark gray patch they will get sadder and if they walk over a white patch, then they will get happier. As teens wander, there are

random deaths that are caused by accidents such as car accidents or fatal fires. When these deaths occur, then a pink patch appears where the teen died. Each member of the group of friends that the teen belonged to has their emotion-level lowered as a result. When a teen's emotion-level drops to -45 , then that teen will attempt suicide. If the teen is successful, a purple patch will appear where the suicide happened. Each member of the group of friends of that the teen belonged to has their emotion-level lowered as a result. As the teens wander around, the counselors do as well. They check the eight spaces surrounding them for any depressed teens. If there is a depressed teen in it's area, then the counselor will try to counsel the teen.

Assumptions:

- We assume that depressed teens (emotion-level = -45) will always commit suicide.
- We assume that counselors counsel the first teen they encounter who is depressed.
- We assume that once a death or suicide occurs at a location, the location forever has a negative effect on teens. (Time does not heal depressing places.)
- We assume that there is no addition to the population of teens, only the removal of teens.
- We assume that counselors never get depressed.

Simplification:

All agents and behaviors used in this model are simplifications of the actual world. With this simulation we aim to understand some of the dynamics of suicide not to simulate the real world of teens.

The Results:

We ran several experiments using our simulation. Each simulation experiment was run for 1000 cycles or loops.

In the first set of experiments we aimed to determine the impact of different percentages of happy and sad spaces in the teens' environment. We decided to set up a basic case in which # of teens = 400, odds of accidental death = 1 in 4000, and number of counselors = 2% of the teen population.

Table 1: The % of sad zones = 15 and the % of happy zones varied, the results were:

	Accidental Deaths	Suicides
% of happy zones = 10	72	101
% of happy zones = 20	98	18
% of happy zones = 30	91	1
% of happy zones = 40	99	0
% of happy zones = 50	109	0
% of happy zones = 60	88	0
% of happy zones = 70	77	0
% of happy zones = 80	79	0

Table 2: The % of happy zones = 10 and the % of sad zones varied, the results were:

	Accidental Deaths	Suicides
% of sad zones = 15	72	101
% of sad zones = 25	69	144
% of sad zones = 35	73	143
% of sad zones = 45	61	155
% of sad zones = 55	59	154
% of sad zones = 65	76	157
% of sad zones = 75	58	151
% of sad zones = 85	59	118

Table 3: The % of happy zones = 10, the % of sad zones = 15, and the density of counselors varied, the results were:

	Accidental Deaths	Suicides
Density of Couns. = 2	72	101
Density of Couns. = 10	84	33
Density of Couns. = 20	90	14
Density of Couns. = 30	92	1

We also monitored and graphed the population of each friend group over time and the average emotion-level of each friend group over time. (See Appendix B).

Conclusion:

When we analyzed the results of our experiments we found that

- 1) as the % of happy zones increased, the number of suicides decreased.
- 2) as the % of sad zones increased, the number of suicides increased.
(except in the case where % of sad zones = 85.)
- 3) as the density of counselors increased, the number of suicides decreased.

In Charts 1- 4 (see appendix B) the graphs of the average emotion-level of the friend groups clearly shows a descending of emotion-level when the density of counselors = 2%. While at higher densities of counselors, there seems to be more “resilience”; the average emotion-level of the friend groups rebounded after suffering a depressing period.

When watching a simulation and looking at the graphs of the average emotion-level of the friend groups, it became apparent that an accidental death or a large dose of negative events triggered an initial suicide. If no intervention or positive events occurred after the suicide, other teens often followed in committing suicide. This rippling effect of suicide through a teen community is similar to that which occurred in Pojoaque, NM over the past 12 months.

An unexpected pattern sometimes emerges when a suicide occurs in an otherwise happy group of teens. Instead of a decrease of the average emotion-level for the group, there is a rise in the average emotion-level because the depressed friend has been removed from the average.

Significant Original Achievement:

Our most significant original achievement was to develop a mathematical agent-based model of the psychological phenomena of teen suicide and depression. In an Internet search for similar research, we did not find any other agent-based models of the spread of suicide in teens. We think we chose a creative way to look at the problem of teen suicide. Our model can be used as a test-bed for trying out multiple scenarios.

Future steps:

To take this model even further, we would aim to make it more closely resemble the real world. We could put in more teen attributes or factors in suicide such as

- additional risk factors for suicide: genetic predisposition, access to firearms, alcohol or drug use, relative of a suicide, impulsivity, etc.
- protective factors: self-esteem, feelings of self-worth, etc.

In addition, we could take into account that some teens are loners and some teens belong to multiple friend groups. Lastly, we could make the environment more closely resemble a teen’s environment or school.

References

Borowsky, I. W., Ireland, M., Resnick, M. D. "Adolescent Suicide Attempts: Risks and Protectors". *Pediatrics* 107 (3), 2001, pgs, 485-493.

Constable, Anne, "Statistics show many teen in the state have thought about suicide", *Santa Fe New Mexican*, February 29, 2004. (online)

Kusel, Denise, "Discussing suicide helps lift the veil of shame", *Santa Fe New Mexican*, January 28, 2004, pages B1 and B5.

Kusel, Denise, "Suicide prevention a complex issue that needs a voice", *Santa Fe New Mexican*. February 2004.

Kusel, Denise, "Look beyond the myths of suicide and find help now", *Santa Fe New Mexican*. February 2004.

La Cerva, Victor, "Let Peace Begin With Us: The Problem of Violence in New Mexico" volume IV, May 2002. New Mexico Department of Health.

La Cerva, Victor. Personal communication.

Stone, Marissa, "Suicide Watch", *Santa Fe New Mexican*, February 22, 2004. Pages A1 and A5

Stone, Marissa, "The sorrows of suicide", *Santa Fe New Mexican*, February 29, 2004. (online)

Thompson, E.A., Eggert, L. L., Randell, B. P., et al. "Evaluation of indicated suicide risk prevention approaches for potential high school dropouts", *American Journal of Public Health*, 2001, 91(5): pgs. 742-752.

Websites:

www.freeessays.cc Teen social issues

www.mentalhealth.org/suicideprevention/

www.surgeongeneral.gov

Acknowledgements

We would like to thank Irene Lee for guiding us through the use of Starlogo. We would like to thank Paige Prescott for being our Sponsoring teacher and constantly keeping us on track. Thanks to Victor LaCerva MD who met with us and gave us helpful material and statistics to help us make our project realistic. We would also like to thank Paquita Hernandez for allowing us to work with the Starlogo Mentorship through Monte del Sol Charter School.

Appendix A

The Starlogo code

```
:: Observer procedures
```

```
globals [  
  thisfriend      ; when a teen dies, it's friend group number is stored in this global variable  
  loopcount      ; this global counts how many times the whole 'go' loop has been executed  
  numrandomdeaths ; this is the number of accidental deaths that have occurred  
  numsuicides    ; this is the number of suicides that have occurred  
]
```

```
to setup      ; called by the Setup button  
  ca          ; clear the screen of all previous graphics, turtles, etc.  
  createpatches ; create patches as happy and sad patches  
  createteens ; create the teens  
  createcouns ; creates the counselors  
  clearplots  ; clears the plots  
  setcolors   ; set teens' colors based on their friend groups  
  set thisfriend -1 ; set thisfriend global is set to -1, so no one is affected by it at first  
  set loopcount 0 ; set loopcount to 0  
  set numrandomdeaths 0 ; set number of accidental deaths to 0  
  set numsuicides 0 ; set number of suicides to 0  
  outputsettings ; output the setting of the sliders to the output window  
end
```

```
to createpatches  
  let [:numsadpatches ((screen-height * screen-width) * (%sadplaces / 100))]  
    ; compute the number of sad patches  
  let [:numhappypatches ((screen-height * screen-width) * (%happyplaces / 100))]  
    ; compute the number of happy patches  
  crt 1 ; creates a single turtle  
  repeat :numsadpatches  
    [ask-turtles [placeUniqueSad]] ; place a sad patch on an unused spot  
  repeat :numhappypatches  
    [ask-turtles [placeUniqueHappy]] ; place a happy patch on an unused spot  
  ask-turtles [die] ; then turtle dies  
end
```

```
to createteens  
  create-and-do numteens [ ; create number of turtles specified with the slider numteens  
    set breed teens ; set the new turtles to be teens  
    setxy random screen-width random screen-height ; place them randomly on the screen  
    set emotion-level 0 ; set emotion-level to the neutral = 0  
    set moodiness ((random 10) + 1) ; set moodiness to a random number between 1 and 10  
    set friend-number random 5 ; they're are randomly divided into 5 friend groups  
    set gender random 2 ; set gender to either 0 (male), or 1 (female)  
  ]  
end
```

```

to createcouns
  create-and-do %couns * numteens / 100 [      ; create number of turtles specified using density slider
    set breed couns                          ; set the new turtles to be counselors
    setxy random screen-width random screen-height ; place them randomly on the screen
    set shape couns-shape                    ; set their shape to look like little nurses
    set friend-number 5                      ; set friend-number to 5 so will not interfere with teens
  ]
end

```

```

to setcolors
  ask-turtles [
    if friend-number = 0 [setcolor 25 set friend-color 25] ; friend-group 0 is set to orange
    if friend-number = 1 [setcolor 45 set friend-color 45] ; friend-group 1 is set to yellow
    if friend-number = 2 [setcolor 65 set friend-color 65] ; friend-group 2 is set to lime
    if friend-number = 3 [setcolor 85 set friend-color 85] ; friend-group 3 is set to aqua
    if friend-number = 4 [setcolor 105 set friend-color 105] ; friend-group 4 is set to blue
  ]
end

```

```

to outputsettings
  type "number of teens " print numteens
  type "density of councelors: " print %couns
  type "odds of events: " print event
  type "%happyplaces: " print %happyplaces
  type "%sadplaces: " print %sadplaces
end

```

```

to observerGo      ; called by the Go button
  ask-turtles [go] ; each loop starts here
  set loopcount loopcount + 1 ; the loopcount is increased by one
  output-data      ; write current data to the output window
  if loopcount = 1000 [stopall] ; if the loopcount is 1000, everything stops.
  if count-teens = 0 [stopall] ; or if all the teens are dead, stop everything
end

```

```

to output-data
  type loopcount type ", " ; current loopcount
  type numrandomdeaths type ", " ; number of random deaths
  type numsuicides type ", " ; number of suicides

  type count-teens-with [friend-number = 0] type ", " ; # teens alive in friend-group 0
  type count-teens-with [friend-number = 1] type ", " ; # teens alive in friend-group 1
  type count-teens-with [friend-number = 2] type ", " ; # teens alive in friend-group 2
  type count-teens-with [friend-number = 3] type ", " ; # teens alive in friend-group 3
  type count-teens-with [friend-number = 4] type ", " ; # teens alive in friend-group 4

  type average-of-teens-with [friend-number = 0] [emotion-level] type ", " ; av emotion-level of group 0
  type average-of-teens-with [friend-number = 1] [emotion-level] type ", " ; av emotion-level of group 1
  type average-of-teens-with [friend-number = 2] [emotion-level] type ", " ; av emotion-level of group 2
  type average-of-teens-with [friend-number = 3] [emotion-level] type ", " ; av emotion-level of group 3
  print average-of-teens-with [friend-number = 4] [emotion-level] ; av emotion-level of group 4
end

```

```

;; Turtle procedures

breeds [
  teens ; the breed of teenagers, the main occupants of Teen World
  couns ; the breed of healing counselors
]

turtles-own [
  emotion-level ; emotional state (between -45 and 45)
  moodiness ; amount of change possible in the emotion-level per step (between 1 and 10)
  friend-number ; group of friends to which the teen belongs
  friend-color ; the color corresponding to the friend-number
  gender ; gender of the teen
]

to go ; basic turtle go loop
  fd 1 ; take one step forward
  ifelse breed = teens ; if a teen, run
    [teensgo] ; teens go loop
    [counsgo] ; else run counselors go loop
end

to teensgo ; the teen go loop
  scale-color friend-color emotion-level -45 45 ; scale teen's color to reflect it's emotion-level
  check-patches ; react to the patches around it
  if emotion-level > 45 [set emotion-level 45] ; cap teen's emotion-level
  if emotion-level < -45 [set emotion-level -45] ; cap teen's emotion-level
  ifelse ((random event) = 1) ; if random number between 0 and event-1 = 1
    [randomdeath] ; teen has an accidental death occurs
    [check-friends-of-death] ; else check if any of its friends have died
  ifelse color = 0 ; if color is black (emotion-level = -45)
    [attempt] ; teen makes a suicide attempt
    [check-friends-of-suicide] ; else check if any of friends have committed suicide
  set thisfriend -1 ; set variable thisfriend to -1
end

to check-patches
  ifelse pc-ahead = 03 ; if patch ahead is gray, decrement emotion-level
    [set emotion-level emotion-level + (-0.5 * (moodiness)) rt random 360]
    [ifelse pc-ahead = 115 ; else if the patch ahead is purple, set emotion-level accordingly
      [set emotion-level emotion-level + (((random 3) - 3) / 3 * (moodiness)) rt random 360]
      [ifelse pc-ahead = 135 ; else if the patch ahead is pink
        [set emotion-level emotion-level + (((random 2) - 2) / 2 * (moodiness)) rt random 360]
        [ifelse pc-ahead = 09 ; else if the patch ahead is white, increment emotion-level
          [set emotion-level emotion-level + (.5 * (moodiness)) rt random 360]
          [rt random 360] ; else just turn a random amount
        ]
      ]
    ]
end

```

```

to check-friends-of-death ; check to see if a friend died in an accident
  if friend-number = thisfriend ; if teen that died is in my friend group
    [set emotion-level emotion-level - (.75 * moodiness)] ; decrement my emotion-level
  end

to check-friends-of-suicide ; check to see if a friend committed suicide
  if friend-number = thisfriend ; if teen that committed suicide is a friend
    [set emotion-level emotion-level - (moodiness)] ; subtract my moodiness from my emotion-level
  end

to randomdeath ; accidental death
  stamp 135 ; stamps a pink spot to mark its place of death
  set thisfriend friend-number ; set global thisfriend to friend group number
  set numrandomdeaths numrandomdeaths + 1 ; increment the number of random deaths
  die ; then die
end

to attempt ; attempt suicide
  ifelse gender = 0 ; if the teen's gender 0, it's male, and 1 is female
    [if (random 5) = 0 [suicide]] ; males have 1 in 5 chance of successful suicide attempt
    [if (random 20) = 0 [suicide]] ; females have a 1 in 20 chance of successful suicide attempt
  end

to suicide ; successful suicide attempt
  stamp 115 ; stamps a purple spot to mark its place of suicide
  set thisfriend friend-number ; set global thisfriend to friend group number
  set numsuicides numsuicides + 1 ; increment the number of suicides
  die ; then die
end

to counsgo ; the counselor go loop
  teencheck ; look for teen to counsel
  rt random 360 ; then they turn a random amount
end

to teencheck
  let [:currentdir ((random 8) * 45)] ; set :currentdir two random 45 degree interval
  repeat 8 ; repeats the following 8 times, once for each direction:
    [
      seth :currentdir ; set heading to the :currentdir
      if ((count-teens-towards 0 1) > 0) and (emotion-level-towards 0 1) < 0
        [
          ; if depressed teen is in that direction
          grab (one-of-teens-towards 0 1) ; grab it and set it's emotion-level
          [ set emotion-level-of partner emotion-level + (((random 4) - 1) * (moodiness)) ]
          stop ; breaks out of the above repeat loop
        ]
      set :currentdir :currentdir + 45 ; increment :currentdir by +45, and check again
    ]
end

```



```

to placeUniqueHappy          ; place a happy patch down where no other happy or sad patch exists
  if pc = black              ; if the spot is unused or neutral, make it white or happy
    [stamp 9 stop]          ; break out of recursion
    setxy random screen-width random screen-height ; get another random x-y location
    placeUniqueHappy        ; try again to find an unused or neutral spot
  end

```

```

to placeUniqueSad           ; place a sad patch down where no other happy or sad patch exists
  if pc = black             ; if the spot is unused or neutral, make it dark gray or sad
    [stamp 3 stop]         ; break out of recursion
    setxy random screen-height random screen-width ; get another random x-y location
    placeUniqueSad         ; try again to find an unused or neutral spot
  end

```

Appendix B

Charts 1 – 4: In this set of experiments, the density of counselors varied while the other variables stayed constant. (# teens = 400, odds of accidental death = 1 in 4000, % of happy zones = 10, and % of sad zones = 15.)




Chart 1.




Chart 2.




Chart 3.





Chart 4.

Sample Screen



Sample Data Output

number of teens 400
 density of counselors: 0
 odds of events: 4000
 %happyplaces: 10
 %sadplaces: 15

1	0	0	66	93	77	70	94	-0.56061	-0.17742	-0.16234	-0.12143	0.015957
2	0	0	66	93	77	70	94	-0.81818	-0.2957	-0.41558	-0.14286	-0.23936
3	1	0	66	92	77	70	94	-1.15909	-0.22826	-0.88961	-0.37857	-0.32447
4	1	0	66	92	77	70	94	-1.09091	-0.54891	-0.82468	-0.55	-0.47872
5	1	0	66	92	77	70	94	-0.95455	-0.51087	-1.02597	-0.88571	-0.56915
6	1	0	66	92	77	70	94	-1.07576	-0.68478	-1.33766	-1.04286	-1.06383
7	1	0	66	92	77	70	94	-1.5303	-0.83696	-1.81169	-1.32857	-1.32979
8	1	0	66	92	77	70	94	-1.90152	-1.04891	-1.86364	-1.32143	-1.60638
9	1	0	66	92	77	70	94	-2.05303	-1.17935	-1.86364	-0.94286	-1.81383
10	1	0	66	92	77	70	94	-2.30303	-1.29891	-1.67532	-0.9	-2.09043
11	1	0	66	92	77	70	94	-2.10606	-1.40217	-1.75325	-1.37857	-2.3617
12	1	0	66	92	77	70	94	-2.37879	-1.5	-1.92857	-1.75714	-2.44149
13	1	0	66	92	77	70	94	-2.80303	-1.6087	-1.96753	-2	-2.5266